15–150: Principles of Functional Programming

Lazy Functional Programming

Michael Erdmann*
Spring 2018

1 Topics

• Lazy programming (in a call-by-value language)
• Delaying computation
• Demand-driven manipulation of infinite data

2 Introduction

SML is a call-by-value language: functions evaluate their arguments. So an attempt to build an infinite list, such as the following code fragment, is doomed to failure (actually, will loop forever):

(* repeat : 'a list -> 'a list *)
fun repeat L = L @ repeat L

Even though SML tells us this function is well-typed, if we apply it to a list we won’t get a value. For example

repeat [0] ==> [0] @ repeat [0]
  ==> [0] @ ([0] @ repeat [0])
  ==> ... forever

Although this is a silly example, it illustrates a problem. We may need to build and manipulate infinite data structures, and deal with such data in a demand-driven manner: just evaluate as much as we need, while leaving the rest for later if we need it. The call-by-value behavior of functions needs to be side-stepped somehow, if we want to prevent eager evaluation of the rest of the data.

In fact we can simply use functions themselves to delay computation. Function values (e.g., lambda expressions) are values: SML doesn’t evaluate inside a function body until, if and when, the function is applied to an argument value.

* Adapted from a document by Stephen Brookes.
3 Representing Infinite Data

In a functional programming language, one way to represent an infinite list of data of some type \( t \) is as a function \( \text{int} \rightarrow t \). However, this kind of representation is not well-tailored for lazy demand-driven computation, particularly when future data may depend on external input. Moreover, such a representation may create a lot of tedious index manipulation.

Today we will introduce another representation for potentially infinite lists, which we call \textit{streams}. We will implement operations on streams that facilitate solving non-trivial demand-driven problems.

The main idea is that a \( t \text{ stream} \) of values of type \( t \) is really a delayed computation. We delay a computation by making it be the body of a function known as a \textit{suspension}, sometimes also called a \textit{thunk}. When the function is called on its argument, it computes the \textit{head} of the stream, along with another suspension for computing the next element of the stream, and so forth.

\textbf{Definition:} A \textit{suspension} of a type \( \tau \) is a function of type \( \text{unit} \rightarrow \tau \). We say a suspension is \textit{forced} when it is applied to argument \( () \).

The reason we care about suspensions is that we want to delay evaluation. In an eager language like SML, writing \( e \) (e.g., in the REPL) causes evaluation of \( e \). Writing \( (\text{fn} () \Rightarrow e) \) prevents immediate evaluation of \( e \) since lambda expressions are already values. Only when forced will the suspension cause evaluation of \( e \).

\textbf{Streams:} We now define streams as follows:

\begin{verbatim}
datatype 'a stream = Stream of unit -> 'a front
and 'a front = Empty | Cons of 'a * 'a stream
\end{verbatim}

Observe that for any type \( t \), a \( t \text{ stream} \) is basically a suspension of a \( t \text{ front} \), which looks much like a \( t \text{ list} \), only now as deferred computation. Specifically, a (nonempty) \( t \text{ front} \) is a “cons” of an element of type \( t \) and another \( t \text{ stream} \). (Aside: Alternatively, we could have defined \( t \text{ stream} = \text{unit} \rightarrow t \text{ front} \) without a \texttt{Stream} constructor. We chose the current design for visual clarity when debugging code.)

We need a function for turning a suspension of a front into a stream, and for exposing the first element of a stream by forcing the underlying suspension:

\begin{verbatim}
(* delay : (unit -> 'a front) -> 'a stream *)
fun delay d = Stream(d)

(* expose : 'a stream -> 'a front *)
fun expose (Stream(d)) = d()
\end{verbatim}

We also have an empty stream, which is essentially just a suspension of an empty front:

\begin{verbatim}
val empty : 'a stream = Stream(fn () => Empty)
\end{verbatim}

Finally, we have a function for adding an element to the beginning of a stream, returning the resulting stream:

\begin{verbatim}
(* cons : 'a * 'a stream -> 'a stream *)
fun cons (x, s) = Stream(fn () => Cons(x, s))
\end{verbatim}
3.1 Examples

Here are some examples that construct values of type int stream:

(* zeros : unit -> int front *)
fun zeros () = Cons(0, delay zeros)

(* Zeros : int stream *)
val Zeros = delay zeros

Zeros is an infinite stream of 0s.

(* ones : unit -> int front *)
fun ones () = Cons(1, delay ones)

(* Ones : int stream *)
val Ones = delay ones

Ones is an infinite stream of 1s.

(* streamify : 'a list -> 'a stream *)
fun streamify [] = empty
| streamify (x::L) = cons(x, streamify L)

The function streamify turns a list into a finite stream with the same elements as in the list (and in the same order).

(* natsFrom : int -> (unit -> int front) *)
fun natsFrom n () = Cons(n, delay (natsFrom (n+1)))

(* Nats : int stream *)
val Nats = delay (natsFrom 0)

The function natsFrom constructs a suspension of a front, that represents the natural numbers from some number upward. Nats is the stream of all natural numbers, in ascending order.

(* lazyappend : 'a list * 'a stream -> 'a stream *)
fun lazyappend ([ ], s) = s
| lazyappend (x::xs, s) = cons(x, lazyappend(xs,s))

The function lazyappend appends a list to a stream, returning a stream of all the list elements followed by all the original stream elements.

(* repeat : 'a list -> 'a stream *)
(* REQUIRES: xs is not nil *)
(* ENSURES: take(repeat xs, n) returns a value for all n >= 0. *)
(* (see definition of take below) *)
fun repeat xs = lazyappend(xs, delay(fn () => expose(repeat xs)))

The function repeat expects a list and returns a stream that represents infinite appends of the list to itself. For example, repeat [1,2,3] is the stream consisting of 1,2,3,1,2,3,..., repeating forever.

What happens if we evaluate repeat [ ]?
What happens if we evaluate expose(repeat [ ])?
3.2 Viewing a Stream

Here is a useful function that shows us a “view” of a stream: we can look at the first $n$ items in the stream, as an ordinary finite list.

\[
\text{take : 'a stream * int} \rightarrow \text{'}a\text{ list}
\]

\[
\text{take' : 'a front * int} \rightarrow \text{'}a\text{ list}
\]

REQUIRES: $n \geq 0$

ENSURES: \text{take}(s,n) returns a list of the first $n$ elements of $s$ in order, unless $s$ has fewer than $n$ elements in which case \text{take}(s,n) raises Subscript. If computation of any of these $n$ elements loops forever or raises an exception, then so will \text{take}(s,n).

\[
\begin{align*}
\text{fun take (s, 0) = nil} \\
| \text{take (s, n) = take'} (\text{expose s, n}) \\
\text{and take'} (\text{Empty, _}) = \text{raise Subscript} \\
| \text{take'} (\text{Cons(x,s), n} = x::\text{take}(s, n-1)
\end{align*}
\]

Example: \text{take(Nats, 5) = [0,1,2,3,4]}.  

Note: “taking” the first few elements of a stream forces the first few suspensions to be applied to their argument ($) to extract data from the tail of the stream. This is all purely functional, without side-effect. If we evaluate \text{take(Nats, 2) after take(Nats, 5)} we will get the result [0,1], not [5,6]. The first take doesn’t change the value of the stream.

Observe: The mutual recursion between \text{take} and \text{take'} is a useful template for writing stream functions: one function specializes to streams, the other to fronts, and they hand data back and forth between each other.

3.3 Example: Reals as Infinite Streams of Integers

A stream of integers could be used to represent a real number. We could use the head of the stream to represent the integer part of the real number, and then the rest of the stream to represent the decimal digits, with the most significant digit (the tenths part) first, then the hundredths part, and so on. According to this representation, \text{Ones} represents the real number whose decimal expansion is 1.11111....

\[
\begin{align*}
\text{val X = lazyappend ([5, 8], repeat [1,4,4])} \\
(* X represents the real number equal to 3227/555 *) \\
(* This number has decimal expansion 5.8144144144. . . *)
\end{align*}
\]

\[
\begin{align*}
\text{val Y = repeat [0,1,2,3,4,5,6,7,9]} \\
(* Y represents 10/81 *) \\
(* The decimal expansion for 10/81 is 0.12345679012345679. . . *)
\end{align*}
\]

\[
\begin{align*}
\text{val Z = repeat [0,9]} \\
(* Z represents 10/11 *) \\
(* The decimal expansion for 10/11 is 0.909090. . . *)
\end{align*}
\]
3.4 Higher-Order Functions on Streams

The following functions appear within the Stream structure appearing in the code associated with today’s lecture, as do the types and key functions presented so far. We omit the “Stream.” prefix except when needed to distinguish from similarly named list functions.

- **map**: (‘a -> ’b) -> 'a stream -> 'b stream
  
  ```
  fun map f s = delay (fn () => map' f (expose s))
  and map' f (Empty) = Empty
  | map' f (Cons(x,s)) = Cons(f(x), map f s)
  ```

  If \( f : t_1 \rightarrow t_2 \) is total, and \( S : t_1 \) stream has elements \( d_0, d_1, ..., d_n, ... \), then \( \text{map} \ f \ S \) has type \( t_2 \) stream, and elements \( f(d_0), f(d_1), ..., f(d_n), ... \).

  Again, notice the mutual recursion between a function that operates on streams and one that operates on fronts. Also, observe that \( \text{map} \ f \ s \) returns a delayed computation. Let’s call it \( \text{ms} \). Not until someone, if ever, evaluates \( \text{expose}(\text{ms}) \) does any actual mapping occur, and then only to the first stream element, with further mapping of the rest of the stream itself delayed until requested via an \( \text{expose} \) of the rest of the mapped stream. Etc.

- **zip**: 'a stream * 'b stream -> ('a * 'b) stream
  
  ```
  fun zip (s1, s2) = delay (fn () => zip'(expose s1, expose s2))
  and zip' (_, Empty) = Empty
  | zip' (Empty, _) = Empty
  | zip' (Cons(x, s1), Cons(y, s2)) = Cons((x, y), zip(s1, s2))
  ```

  If \( A : t_1 \) stream has elements \( a_0, a_1, ..., a_n, ... \)
  
  and \( B : t_2 \) stream has elements \( b_0, b_1, ..., b_n, ... \),
  
  then \( \text{zip}(A, B) \) evaluates to a value \( Z : (t_1 \ast t_2) \) stream, with elements \( (a_0, b_0), (a_1, b_1), ..., (a_n, b_n), ... \).

- **filter**: (‘a -> bool) -> 'a stream -> 'a stream
  
  ```
  fun filter p s = delay (fn () => filter' p (expose s))
  and filter' p (Empty) = Empty
  | filter' p (Cons(x,s)) = if p(x) then Cons(x, filter p s) else filter' p (expose s)
  ```

  If \( S : t \) filter and \( p : t \rightarrow \) bool is total, \( \text{filter} \ p \ S \) computes the stream of elements of \( S \) that satisfy \( p \).
If \( S \) has an element satisfying \( p \), and \( p \) is total, then \( \text{expose}(\text{filter } p \ S) \) returns a front \( \text{Cons}(x, s) \), where \( x \) is the first element of \( S \) that satisfies \( p \) and \( s \) is a stream representing the remaining elements of \( S \) satisfying \( p \).

What happens if we evaluate \( \text{expose}(\text{filter } (\text{fn } _\Rightarrow \text{false}) \ Ones) \)?

It is natural to see the first two functions as “extensions” of the corresponding functions on ordinary lists, because one can prove the following results:

For all non-negative \( n \), and all suitably typed values \( f, S \) and \( T \), with \( f \) total:

\[
\begin{align*}
\text{take}(\text{Stream.map } f \ S, n) &\equiv \text{List.map } f (\text{take}(S, n)) \\
\text{take}(\text{Stream.zip } (S, T), n) &\equiv \text{List.zip } (\text{take}(S, n), \text{take}(T, n))
\end{align*}
\]

### 3.5 Equivalence of Streams

When should we say that two stream expressions are extensionally equivalent? In what sense are \( \text{Stream.map } (f \circ g) \ S \)

\( \text{Stream.map } f \ (\text{Stream.map } g \ S) \)

extensionally equivalent, assuming as usual that \( f \) and \( g \) are total functions whose types allow them to be composed?

A notion of equivalence that makes sense for streams is that stream values \( X \) and \( Y \) are \textit{extensionally equivalent} iff, for all \( n \geq 0 \), \( \text{take}(X, n) \) and \( \text{take}(Y, n) \) are equivalent. This definition also allows for the possibility that an attempt to extract an element may loop forever (as in the \text{filter} example when there is no element satisfying the predicate). So two streams whose first 42 elements are equivalent and whose 43rd elements result in infinite loops when exposed are also regarded as equivalent.

It is impossible to implement an equality-check for infinite streams, even for \texttt{int streams}. Unlike ordinary lists of integers, we cannot use the built-in \texttt{=} operator of SML to check for equality. Instead, one generally must establish stream equivalence by proof.

### 3.6 Lazy Fusion Equivalences

Given the notion of stream equivalence above, the following \textit{lazy fusion} results are provable:

For all types \( t_1, t_2, t_3 \), all total functions \( f : t_2 \rightarrow t_3 \) and \( g : t_1 \rightarrow t_2 \), and all values \( S : t_1 \text{ stream} \),

\[
\text{Stream.map } (f \circ g) \ S \equiv \text{Stream.map } f \ (\text{Stream.map } g \ S).
\]

The proof proceeds by showing that (under the given assumptions) for all \( n \geq 0 \),

\[
\text{take}(\text{Stream.map } (f \circ g) \ S, n) \equiv \text{take}(\text{Stream.map } f \ (\text{Stream.map } g \ S), n).
\]

In fact, we can appeal to the earlier result linking \text{Stream.map} and \text{List.map}, and the familiar fusion property of \text{List.map}, i.e., that

\[
\text{List.map } (f \circ g) \ (\text{take}(S, n)) \equiv \text{List.map } f \ (\text{List.map } g \ (\text{take}(S, n)))
\]

to derive the lazy fusion equivalence stated above.

Note that we can paraphrase this lazy fusion equivalence, taking advantage of extensionality. The following is an equivalent statement to the one given above:

For all types \( t_1, t_2, t_3 \), and all total functions \( f : t_2 \rightarrow t_3 \) and \( g : t_1 \rightarrow t_2 \),

\[
\text{Stream.map } (f \circ g) \equiv (\text{Stream.map } f) \circ (\text{Stream.map } g).
\]
4 More Examples

This section presents two examples illustrating how to program with streams. The art is in the design of suspensions (often using recursion), and making the control flow sufficiently lazy that we never prematurely reveal the head-tail (i.e., front) structure of part of a stream until we need to.

4.1 Streams of Factorials

Useful facts about factorials:

- $0! = 1$
- $k! \cdot (k + 1) = (k + 1)!$ when $k \geq 0$.

The following lazy way to generate factorials is based on these useful facts. The first fact tells us that the first element of the stream should be 1. The second fact tells us that we can obtain the later factorials by zipping each integer $k + 1$ with the factorial of its predecessor and multiplying. So we embed a recipe for generating factorials starting from $0!$, by zipping and multiplying, in the tail suspension.

(* factorials : unit -> int front *)
fun factorials () = Cons(1, map (op * ) (zip(delay factorials, delay(natsFrom 1))))

(* Factorials : int stream *)
val Factorials = delay factorials

Factorials is a stream consisting of all factorials $n!$, for $n \geq 0$.
In other words, for all non-negative $n$, $\text{take(Factorials, } n+1\text{)}$ evaluates to the list

$$[0!, 1!, \ldots, n!]$$

$\text{take(Factorials, 10)} \Rightarrow [1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880]$. 
4.2 Stream of Prime Numbers

This code is inspired by the Sieve of Eratosthenes, an ancient algorithm for enumerating the primes.

(* notDivides : int -> int -> bool
   REQUIRES: true
   ENSURES: notDivides p q ==> true iff p does not divide q, i.e.,
             q is not a multiple of p. *)

fun notDivides p q = (q mod p <> 0)

(*
   sieve : int stream -> int stream
   sieve' : int front -> int front
   *)

fun sieve s = delay (fn () => sieve' (expose s))
and sieve' (Empty) = Empty
| sieve' (Cons(p, s)) = Cons(p, sieve (filter (notDivides p) s))

val primes = sieve(natsFrom 2)

primes is a stream representing all prime numbers, in ascending order. Here are the first 400: